Newsgroup: it.comp.lang.visual-basic
Titolo: ShutDown di windows 2000

Vorrei far spegnere un PC con windows 2000 con VB6 SP5.
Ho provato sia con InitiateSystemShutdown che con ExitWindowsEx,
entrambi hanno lo stesso effetto ossia mi chiudono tutte le applicazioni
programmi caricate fanno il log off ma alla fine esce scritto:
ora e' possibile spegnere il computer.
Preche' non si spegne da solo?
Ciao Pasquale

 Ciao Pasquale
ciaociao

 
Ci sono stati parecchi thread anche recenti sull'argomento: 
groups.ggogle.com
ciao
	maxx
-- 
NOSPAM: Rimuovere i trattini dallo username!
Cut hyphens from my username!

 maxx
ciaociao

 Vorrei far spegnere un PC con windows 2000 con VB6 SP5.
non è farina del mio sacco...l'ho trovata grazie al newsgroup
bye
    Huntercity
Option Explicit
Private Const EWX_LogOff As Long = 0
Private Const EWX_SHUTDOWN As Long = 1
Private Const EWX_REBOOT As Long = 2
Private Const EWX_FORCE As Long = 4
Private Const EWX_POWEROFF As Long = 8
'The ExitWindowsEx function either logs off, shuts down, or shuts
'down and restarts the system.
Private Declare Function ExitWindowsEx Lib "user32" _
(ByVal dwOptions As Long, _
ByVal dwReserved As Long) As Long
'The GetLastError function returns the calling thread's Last -Error
'code value. The last-error code is maintained on a per-thread basis.
'Multiple threads do not overwrite each other's last-error code.
Private Declare Function GetLastError Lib "kernel32" () As Long
Private Const mlngWindows95 = 0
Private Const mlngWindowsNT = 1
Public glngWhichWindows32 As Long
'The GetVersion function returns the operating system in use.
Private Declare Function GetVersion Lib "kernel32" () As Long
Private Type LUID
UsedPart As Long
IgnoredForNowHigh32BitPart As Long
End Type
Private Type LUID_AND_ATTRIBUTES
TheLuid As LUID
Attributes As Long
End Type
Private Type TOKEN_PRIVILEGES
PrivilegeCount As Long
TheLuid As LUID
Attributes As Long
End Type
'The GetCurrentProcess function returns a pseudohandle for the
'current process.
Private Declare Function GetCurrentProcess Lib "kernel32" () As Long
'The OpenProcessToken function opens the access token associated with
'a process.
Private Declare Function OpenProcessToken Lib "advapi32" _
(ByVal ProcessHandle As Long, _
ByVal DesiredAccess As Long, _
TokenHandle As Long) As Long
'The LookupPrivilegeValue function retrieves the locally Unique
'identifier (LUID) used on a specified system to locally represent
'the specified privilege name.
Private Declare Function LookupPrivilegeValue Lib "advapi32" _
Alias "LookupPrivilegeValueA" _
(ByVal lpSystemName As String, _
ByVal lpName As String, _
lpLuid As LUID) As Long
'The AdjustTokenPrivileges function enables or disables privileges
'in the specified access token. Enabling or disabling privileges
'in an access token requires TOKEN_ADJUST_PRIVILEGES access.
Private Declare Function AdjustTokenPrivileges Lib "advapi32" _
(ByVal TokenHandle As Long, _
ByVal DisableAllPrivileges As Long, _
NewState As TOKEN_PRIVILEGES, _
ByVal BufferLength As Long, _
PreviousState As TOKEN_PRIVILEGES, _
ReturnLength As Long) As Long
Private Declare Sub SetLastError Lib "kernel32" _
(ByVal dwErrCode As Long)
Private Sub AdjustToken()
'********************************************************************
'* This procedure sets the proper privileges to allow a Log off Or a
'* shut down to occur under Windows NT.
'********************************************************************
Const TOKEN_ADJUST_PRIVILEGES = &H20
Const TOKEN_QUERY = &H8
Const SE_PRIVILEGE_ENABLED = &H2
Dim hdlProcessHandle As Long
Dim hdlTokenHandle As Long
Dim tmpLuid As LUID
Dim tkp As TOKEN_PRIVILEGES
Dim tkpNewButIgnored As TOKEN_PRIVILEGES
Dim lBufferNeeded As Long
'Set the error code of the last thread to zero using the
'SetLast Error function. Do this so that the GetLastError
'function does not return a value other than zero for no
'apparent reason.
SetLastError 0
'Use the GetCurrentProcess function to set the hdlProcessHandle variable.
hdlProcessHandle = GetCurrentProcess()
If GetLastError <> 0 Then
    systemMessage.AddItem "GetCurrentProcess Error==" & GetLastError
End If
OpenProcessToken hdlProcessHandle, (TOKEN_ADJUST_PRIVILEGES Or TOKEN_QUERY),
hdlTokenHandle
If GetLastError <> 0 Then
    systemMessage.AddItem "OpenProcessToken Error==" & GetLastError
End If
'Get the LUID for shutdown privilege
LookupPrivilegeValue "", "SeShutdownPrivilege", tmpLuid
If GetLastError <> 0 Then
    systemMessage.AddItem "LookupPrivilegeValue Error==" & GetLastError
End If
tkp.PrivilegeCount = 1 ' One privilege to set
tkp.TheLuid = tmpLuid
tkp.Attributes = SE_PRIVILEGE_ENABLED
'Enable the shutdown privilege in the access token of this process
AdjustTokenPrivileges hdlTokenHandle, False, tkp, Len(tkpNewButIgnored), _
tkpNewButIgnored, lBufferNeeded
If GetLastError <> 0 Then
    systemMessage.AddItem "AdjustTokenPrivileges Error==" & GetLastError
End If
End Sub
Private Sub cmdLogoff_Click()
ExitWindowsEx (EWX_LogOff), &HFFFF
systemMessage.AddItem "ExitWindowsEx's GetLastError " & GetLastError
End Sub
Private Sub cmdForceLogoff_Click()
    ExitWindowsEx (EWX_LogOff Or EWX_FORCE), &HFFFF
    systemMessage.AddItem "ExitWindowsEx's GetLastError " & GetLastError
End Sub
Private Sub cmdShutdown_Click()
If glngWhichWindows32 = mlngWindowsNT Then
    AdjustToken
    systemMessage.AddItem "Post-AdjustToken GetLastError " & GetLastError
End If
ExitWindowsEx (EWX_SHUTDOWN), &HFFFF
systemMessage.AddItem "ExitWindowsEx's GetLastError " & GetLastError
End Sub
Private Sub cmdForceShutdown_Click()
If glngWhichWindows32 = mlngWindowsNT Then
    AdjustToken
    systemMessage.AddItem "Post-AdjustToken GetLastError " & GetLastError
End If
ExitWindowsEx (EWX_SHUTDOWN Or EWX_FORCE), &HFFFF
systemMessage.AddItem "ExitWindowsEx's GetLastError " & GetLastError
End Sub
Private Sub Form_Load()
'********************************************************************
'* When the project starts, check the operating system used By
'* calling the GetVersion function.
'********************************************************************
Dim lngVersion As Long
lngVersion = GetVersion()
If ((lngVersion And &H80000000) = 0) Then
glngWhichWindows32 = mlngWindowsNT
systemType.Text = "Windows NT"
Else
glngWhichWindows32 = mlngWindows95
systemType.Text = "Windows 95"
End If
End Sub


Grazie per il codice ma il risultato non cambia alla fine esce sempre la
scirtta:
Ora si puo' spegnere il computer.
A te funziona su Win 2000?
Pasquale

 Pasquale
non ho win2k ma è in pratica un sistema NT come XP
sul mio xp sp1 se clicco su force shutdown mi spegne la macchina
altro ora non posso provare che se non tra spegni e accendi ci faccio notte
bye

 Pasquale
Devi usare EWX_POWEROFF al posto di EWX_SHUTDOWN.
Solo su WinXP SP1 si comportano uguale.
http://msdn.microsoft.com/library/default.asp?url=/library/en-us/sysinfo/base/exitwindowsex.asp
Byee

 Preche' non si spegne da solo?
sui Sistemi Nt da quel che ricordo alla fine il metodo migliore e' usare un
Exe che si chiama "Shotdown2.exe", per il 98 invece si continua con la
Exitwindows
(almeno e' quello che ho capito dagli infiniti post che sono stati scritti
in merito)
-- 
<Gigio2K>
--- 
http://www.g2kweb.it/gigio2k
 ---
Nokia -> I.T.H.N. -> 
http://www.ithn.it/
VB -> I.C.L.VB -> 
http://www.it-lang-vb.net/


OK funziona perfettamente.
Pasquale